

# PRJ 3 LA Crime Dashboard Documentation:

## Data Details:

### Data Source:

<https://catalog.data.gov/dataset/crime-data-from-2020-to-present>


### Data Dictionary:

[https://data.lacity.org/Public-Safety/Crime-Data-from-2020-to-Present/2nrs-mtv8/about\\_data](https://data.lacity.org/Public-Safety/Crime-Data-from-2020-to-Present/2nrs-mtv8/about_data)

## Special Note:


The following documentation assumes DAX is understood by the reader. It is meant to explain the purpose of the measures, not teach the reader DAX.

## Dashboard:



- All of the above **card charts** are generated using DAX measures of aggregate counts. The purpose of each will be described in the measures section.
- **YEAR Month filter** allows for a choice to narrow the granularity of the data and view historical trends on the year and month level.
- The **LA Crime Count Map** depicts the total crimes in a given area of LA with severity measured in both size of bubble and severity of color.

## Data Model:



- Above shows a simple data model utilizing `la_crimes` as the fact table and `crime_weapons`, `date_tables`, and `crimes` as the reference tables.
- `la_crimes` connects to the `date_tables` by connecting the `date` field from `date_tables` to the `date_rptd` field in the `la_crimes` table. Having this date table connected allows for smooth utilization of date data and date time functions in measures and charts.
- `la_crimes` connects to `crime_weapons` using the `weapons_used_code` from both tables. This allows for narrowing in on crime KPIs relating to weapon base crime.
- The `la_crimes` table connects to the `crime` table using the `crime_code` from both tables. This allows for the data and measures to be filtered by crime description creating niche insights.
- All of the tables are connected using a single cross filter direction.

# DAX Measures:

## % Crimes Committed with weapons

% Crimes Committed with Weapon =

```
CALCULATE(  
    COUNT(la_crimes[division_records_num]),  
    NOT(ISBLANK(crime_weapons[weapon_used_code]))  
)  
/  
COUNT(la_crimes[division_records_num])
```

- Using the calculate function the base of weapon based crime data calculations is created.
- The above takes the count of records related to individual crimes with weapon data and divides it by the count of all crimes.
- The output will be the % of crimes committed with a weapon.

## Arrest Rate

Arrest Rate =

```
CALCULATE(  
    COUNT(la_crimes[status_description]),  
    FILTER(la_crimes, la_crimes[status_description] = "Adult Arrest" ||  
la_crimes[status_description] = "Juv Arrest")) /  
COUNT(la_crimes[division_records_num])
```

- Above is the DAX measure that creates the base measure for all Arrest Rate measures. It uses the status\_description field to make sure only crimes that led to an arrest are included in the total that is then divided by the overall crime count.

## Crime Rate per 1000 Residents

Crime Rate per 1000 Residents =

```
SWITCH(  
    TRUE(),  
    SELECTEDVALUE(date_tables[year]) = 2020, (COUNT(la_crimes[crime_code_description]) /  
3890000) * 1000,  
    SELECTEDVALUE(date_tables[year]) = 2021, (COUNT(la_crimes[crime_code_description]) /  
3849000) * 1000,
```

```

SELECTEDVALUE(date_tables[year]) = 2022, (COUNT(la_crimes[crime_code_description]) /
3823000) * 1000,
SELECTEDVALUE(date_tables[year]) = 2023, (COUNT(la_crimes[crime_code_description]) /
3821000) * 1000,
SELECTEDVALUE(date_tables[year]) = 2024, (COUNT(la_crimes[crime_code_description]) /
3803000) * 1000
--- Based on an average of the LA populations YOY decline the 2024 population was able to
be roughly speculated. The populations data was provided by the United States Census Bureau.
)

```

- The above is the base measure for all Crime Rate per 1000 Residents measures. This is done by using a SWITCH() calculation to narrow in on the year and then subsequently give a count of the crimes in that year and divide it by 1000.
- The output will create a fundamental crime KPI to monitor progress in crime rates.

## Property Crime Rate per 1000 Residents:

Property Crime Rate per 1000 Residents =

```

CALCULATE(
    SWITCH(
        TRUE(),
        SELECTEDVALUE(date_tables[year]) = 2020, COUNT(la_crimes[crime_code_description]) /
3890000,
        SELECTEDVALUE(date_tables[year]) = 2021, COUNT(la_crimes[crime_code_description]) /
3849000,
        SELECTEDVALUE(date_tables[year]) = 2022, COUNT(la_crimes[crime_code_description]) /
3823000,
        SELECTEDVALUE(date_tables[year]) = 2023, COUNT(la_crimes[crime_code_description]) /
3821000,
        SELECTEDVALUE(date_tables[year]) = 2024, COUNT(la_crimes[crime_code_description]) /
3803000
    ) * 1000,
    CONTAINSSTRING(crimes[crime_code_description], "homicide")
|| CONTAINSSTRING(crimes[crime_code_description], "burglary")
|| CONTAINSSTRING(crimes[crime_code_description], "theft")
|| CONTAINSSTRING(crimes[crime_code_description], "larceny")
|| CONTAINSSTRING(crimes[crime_code_description], "arson")
)

```

- The above builds off of the Crime Rate per 1000 Residents and then filters to only include crime data that relates to property based crimes using the CONTAINSSTRING() function.

- Narrowing in property crime gives a closer look at the risks property owners face when living in Los Angeles.

## Violent Crime Rate per 1000 Residents

Violent Crime Rate per 1000 Residents =

```
CALCULATE(
    SWITCH(
        TRUE(),
        SELECTEDVALUE(date_tables[year]) = 2020, COUNT(la_crimes[crime_code_description]) /
3890000,
        SELECTEDVALUE(date_tables[year]) = 2021, COUNT(la_crimes[crime_code_description]) /
3849000,
        SELECTEDVALUE(date_tables[year]) = 2022, COUNT(la_crimes[crime_code_description]) /
3823000,
        SELECTEDVALUE(date_tables[year]) = 2023, COUNT(la_crimes[crime_code_description]) /
3821000,
        SELECTEDVALUE(date_tables[year]) = 2024, COUNT(la_crimes[crime_code_description]) /
3803000
    ) * 1000,
    CONTAINSSTRING(crimes[crime_code_description], "homicide")
|| CONTAINSSTRING(crimes[crime_code_description], "assault")
|| CONTAINSSTRING(crimes[crime_code_description], "Robbery")
|| CONTAINSSTRING(crimes[crime_code_description], "rape")
|| CONTAINSSTRING(crimes[crime_code_description], "murder")
|| CONTAINSSTRING(crimes[crime_code_description], "manslaughter")
|| CONTAINSSTRING(crimes[crime_code_description], "kidnapping")
)
```

- The above applies the exact same logic as the property crime DAX measure except the CONTAINSSTRING() is used to include different detail in the data relating to violent crimes.
- Violent Crimes stats are a great way to monitor how well crime is being controlled over all and how safe or dangerous an environment is.

## YOY Measures

Using VAR, CALCULATE(), SAMEPERIODLASTYEAR(), and DIVIDE() the general YOY equation is calculated for Arrest Rate, Crime Count, all of the crime per 1000 residents metrics and weapons based crime metrics. YOY is a useful way to find trends in the data and attempt to anticipate the coming trends based on historical data.

Below is the dax to create each:

## YOY Arrest Rate

YOY Arrest Rate =

```
VAR CurrentYearArrestRate = [Arrest Rate]

VAR PreviousYearArrestRate =
    CALCULATE(
        [Arrest Rate],
        SAMEPERIODLASTYEAR('date_tables'[date])
    )

RETURN
    DIVIDE(
        CurrentYearArrestRate - PreviousYearArrestRate,
        PreviousYearArrestRate,
        2
    )
```

## YOY Crime Count

YOY Crime Count =

```
VAR CurrentYearcrimecount = COUNT(la_crimes[crime_code_description])

VAR PreviousYearcrimecount =
    CALCULATE(
        COUNT(la_crimes[crime_code_description]),
        SAMEPERIODLASTYEAR('date_tables'[date])
    )

RETURN
    DIVIDE(
        CurrentYearcrimecount - PreviousYearcrimecount,
        PreviousYearcrimecount,
        2
    )
```

## YOY Crime Rate per 1000 Residents

YOY Crime Rate per 1000 Residents =

```
VAR currentrate = [Crime Rate per 1000 Residents]
```

```

VAR previousrate =
    CALCULATE(
        [Crime Rate per 1000 Residents],
        SAMEPERIODLASTYEAR('date_tables'[date])
    )
RETURN
    DIVIDE(
        currentrate-previousrate,
        previousrate,
        2
    )

```

## YOY Property Crime Rate per 1000 Residents

YOY Property Crime Rate per 1000 Residents =

```

VAR currentrate = [Property Crime Rate per 1000 Residents]
VAR previousrate =
    CALCULATE(
        [Property Crime Rate per 1000 Residents],
        SAMEPERIODLASTYEAR('date_tables'[date])
    )
RETURN
    DIVIDE(
        currentrate-previousrate,
        previousrate,
        2
    )

```

## YOY Violent Crime Rate per 1000 Residents

YOY Violent Crime Rate per 1000 Residents =

```

VAR currentrate = [Violent Crime Rate per 1000 Residents]
VAR previousrate =
    CALCULATE(
        [Violent Crime Rate per 1000 Residents],
        SAMEPERIODLASTYEAR('date_tables'[date])
    )
RETURN
    DIVIDE(
        currentrate-previousrate,
        previousrate,
        2
    )

```

)

## YOY Weapon Used in Crime

YOY Weapon used in crime =

```
VAR currentrate = [% Crimes Committed with Weapon]
VAR previousrate =
    CALCULATE(
        [% Crimes Committed with Weapon],
        SAMEPERIODLASTYEAR('date_tables'[date])
    )
RETURN
    DIVIDE(
        currentrate-previousrate,
        previousrate,
        2
    )
```

## MOM DAX Measures

MOM measure follows the same logic as the YOY measures except instead of SAMEPERIODLASTYEAR() DATEADD() is used instead. MOM is used to get a short term view of the current and potential trends ahead. Arrest Rate and Crimes Committed with weapons are broken down by MOM via card charts.

## MOM Change Arrest Rate

MOM Change Arrest Rate =

```
VAR CurrentYearArrestRate = [Arrest Rate]

VAR PreviousYearArrestRate =
    CALCULATE(
        [Arrest Rate],
        DATEADD('date_tables'[date], -1, MONTH)
    )

RETURN
    DIVIDE(
        CurrentYearArrestRate - PreviousYearArrestRate,
        PreviousYearArrestRate,
        0
    )
```



## MOM Crimes Committed with Weapon

MOM % Crimes Committed with Weapon =

```
VAR CurrentMonthArrestRate = [% Crimes Committed with Weapon]
```

```
VAR PreviousMonthArrestRate =
```

```
    CALCULATE(  
        [% Crimes Committed with Weapon],  
        DATEADD('date_tables'[date], -1, MONTH)  
    )
```

```
RETURN
```

```
    DIVIDE(  
        CurrentMonthArrestRate - PreviousMonthArrestRate,  
        PreviousMonthArrestRate  
    )
```